AMS-511 Foundations of Quantitative Finance

Fall 2020 — Solutions 03

Robert J. Frey, Research Professor
Stony Brook University, Applied Mathematics and Statistics

Robert.Frey@StonyBrook.edu
http://www.ams.sunysb.edu/~frey

Question 1

NOTE: This question was inadvertently included in both Assignments 02 and 03.

Use FinancialBond[ ] to compute the following:

- A newly issued $10,000 bond with a ten-year term has a 4% coupon rate and pays semi-annually. The current market yield for a ten-year bond of this type is 3.7%. What is the current market price of the bond?
- A $10,000 bond with a five-year term, a 3.7% coupon rate paying semi-annually was issued eight months ago. The current market yield is 3.5%.
  - What is the current price of the bond?
  - What is the accrued interest on the bond?
- Consider a $10,000 semi-annual bond with a twenty-year term and a 4.5% coupon rate. The bond was issued five-years ago and currently trades at a price of $10,500. What is the yield on the bond?

Solution

- A newly issued $10,000 bond with a ten-year term has a 4% coupon rate and pays semi-annually. The current market yield for a ten-year bond of this type is 3.7%. What is the current market price of the bond?

```
In[∘]:= FinancialBond[{"Coupon" → 0.04, "FaceValue" → 10 000, "Maturity" → 10,
          "CouponInterval" → 1/2}, {"InterestRate" → 0.037, "Settlement" → 0}]

Out[∘]= 10 248.9
```

- A $10,000 bond with a five-year term, a 3.7% coupon rate paying semi-annually was issued eight months ago. The current market yield is 3.5%.
  - What is the current price of the bond?
  - What is the accrued interest on the bond?

```
In[•]:= FinancialBond[
         {"Coupon" → 0.037, "FaceValue" → 10 000, "Maturity" → 5, "CouponInterval" → 1/2},
         {"InterestRate" → 0.035, "Settlement" → 8/12}, {"Value", "AccruedInterest"}]
Out[•]= {10 079.4, 61.6667}
```

- Consider a \$10,000 semi-annual bond with a twenty-year term and a 4.5% coupon rate. The bond was issued five-years ago and currently trades at a price of \$10,500. What is the yield on the bond?

```
In[•]:= FindRoot[
         10 500. == FinancialBond[{"Coupon" → 0.045, "FaceValue" → 10 000, "Maturity" → 10,
            "CouponInterval" → 1/2}, {"InterestRate" → y, "Settlement" → 5}], {y, 0.045}]
Out[•]= {y → 0.0340402}
```

Question 2

Consider two assets with mean returns, standard deviations and correlation matrix:

$$\mu = \begin{pmatrix} 0.08 \\ 0.05 \end{pmatrix}$$

$$\sigma = \begin{pmatrix} 0.10 \\ 0.04 \end{pmatrix}$$

$$\mathbf{C} = \begin{pmatrix} 1 & 0.4 \\ 0.4 & 1 \end{pmatrix}$$

$$\mathcal{M}_1 = \min_x \left\{ \frac{1}{2} \mathbf{x}^T \, \Sigma \, \mathbf{x} - \lambda \, \boldsymbol{\mu}^T \, \mathbf{x} \;\middle|\; \mathbf{1}^T \, \mathbf{x} = 1 \right\}$$

- Compute the covariance matrix Σ.

Consider the following portfolio optimization problem $\mathcal{M}_1$ (short positions allowed):

$$\mathcal{M}_1 = \min_x \left\{ \frac{1}{2} \mathbf{x}^T \, \Sigma \, \mathbf{x} - \lambda \, \boldsymbol{\mu}^T \, \mathbf{x} \;\middle|\; \mathbf{1}^T \, \mathbf{x} = 1 \right\}$$

- Compute and plot a mean-variance efficient frontier for a portfolio consisting of these two assets.
- If the risk-free rate $r_f = 0.01$, what is the tangent portfolio?

Consider the following portfolio optimization problem $\mathcal{M}_2$ (no short positions allowed):

$$\mathcal{M}_2 = \min_x \left\{ \frac{1}{2} \mathbf{x}^T \, \Sigma \, \mathbf{x} - \lambda \, \boldsymbol{\mu}^T \, \mathbf{x} \;\middle|\; \mathbf{1}^T \, \mathbf{x} = 1 \wedge x_i \geq 0 \right\}$$

- Compute and plot a mean-variance efficient frontier for a portfolio consisting of these two assets.
- Compare the efficient frontiers of $\mathcal{M}_1$ and $\mathcal{M}_2$.

Solution

In[∘]:=
```
vnMu = {0.08, 0.05};
vnSigma = {0.1, 0.04};
mnCor = {{1., 0.4}, {0.4, 1.}};
```

- Compute the covariance matrix $\mathbf{\Sigma}$.

In[∘]:=
```
mnCovar = KroneckerProduct[vnSigma, vnSigma] mnCor;
```

- Compute and plot a mean-variance efficient frontier for a portfolio consisting of these two assets.

$$\mathcal{M}_1 = \min_x \left\{ \frac{1}{2} \mathbf{x}^T \mathbf{\Sigma} \mathbf{x} - \lambda \boldsymbol{\mu}^T \mathbf{x} \;\middle|\; \mathbf{1}^T \mathbf{x} = 1 \right\}$$

$$\mathbf{x} = \lambda \mathbf{\Sigma}^{-1} \boldsymbol{\mu} + \left(\frac{1 - \lambda \mathbf{1}^T \mathbf{\Sigma}^{-1} \boldsymbol{\mu}}{\mathbf{1}^T \mathbf{\Sigma}^{-1} \mathbf{1}} \right) \mathbf{\Sigma}^{-1} \mathbf{1}$$

In[∘]:=
```
xEffPort[λ_, μ_, Σ_] := Block[
   {mnInvΣ, vnOne},
   mnInvΣ = Inverse[Σ];
   vnOne = Array[1. &, Length[μ]];
   λ mnInvΣ.μ + ((1 - λ vnOne.mnInvΣ.μ)/(vnOne.mnInvΣ.vnOne)) mnInvΣ.vnOne
  ];
```

In[∘]:=
```
mnEffPorts = Table[xEffPort[λ, vnMu, mnCovar], {λ, 0, 0.35, 0.025}]
```

Out[∘]=
```
{{2.27374×10^-17, 1.}, {0.0892857, 0.910714}, {0.178571, 0.821429},
 {0.267857, 0.732143}, {0.357143, 0.642857}, {0.446429, 0.553571},
 {0.535714, 0.464286}, {0.625, 0.375}, {0.714286, 0.285714},
 {0.803571, 0.196429}, {0.892857, 0.107143}, {0.982143, 0.0178571},
 {1.07143, -0.0714286}, {1.16071, -0.160714}, {1.25, -0.25}}
```

In[∘]:=
```
mnEffFront = {Sqrt[#.mnCovar.#], vnMu.#} & /@ mnEffPorts
```

Out[∘]=
```
{{0.04, 0.05}, {0.0408285, 0.0526786}, {0.0432187, 0.0553571},
 {0.0469327, 0.0580357}, {0.0516859, 0.0607143}, {0.0572198, 0.0633929},
 {0.0633302, 0.0660714}, {0.0698659, 0.06875}, {0.0767184, 0.0714286},
 {0.0838099, 0.0741071}, {0.0910847, 0.0767857}, {0.0985022, 0.0794643},
 {0.106032, 0.0821429}, {0.113653, 0.0848214}, {0.121347, 0.0875}}
```

- Compute and plot a mean-variance efficient frontier for a portfolio consisting of these two assets.

$$\mathcal{M}_2 = \min_x \left\{ \frac{1}{2} \mathbf{x}^T \mathbf{\Sigma} \mathbf{x} - \lambda \boldsymbol{\mu}^T \mathbf{x} \;\middle|\; \mathbf{1}^T \mathbf{x} = 1 \wedge x_i \geq 0 \right\}$$

In[∘]:=
```
x = {x1, x2};
```

In[•]:=
```
xEffPortNoShorts[λ_, vnMean_, mnCov_] := x /. Last[
    NMinimize[{x.mnCov.x/2 - λ vnMean.x, And @@ Join[{Total[x] == 1}, # ≥ 0 & /@ x]}, x]];
```

In[•]:=
```
mnEffPortNoShort = Table[
   xEffPortNoShorts[λ, vnMu, mnCovar],
   {λ, 0.0, 0.35, 0.025}
  ]
```

Out[•]=
```
{{1.47094 × 10^-8, 1.}, {0.0892858, 0.910714}, {0.178571, 0.821429},
 {0.267857, 0.732143}, {0.357143, 0.642857}, {0.446429, 0.553571},
 {0.535714, 0.464286}, {0.625, 0.375}, {0.714286, 0.285714}, {0.803571, 0.196429},
 {0.892857, 0.107143}, {0.982143, 0.0178572}, {1., 0.}, {1., 0.}, {1., 0.}}
```

In[•]:=
```
mnEffFrontNoShort = {Sqrt[#.mnCovar.#], vnMu.#} & /@ mnEffPortNoShort
```

Out[•]=
```
{{0.04, 0.05}, {0.0408285, 0.0526786}, {0.0432187, 0.0553571},
 {0.0469327, 0.0580357}, {0.0516859, 0.0607143},
 {0.0572198, 0.0633929}, {0.0633302, 0.0660714}, {0.0698659, 0.06875},
 {0.0767184, 0.0714286}, {0.0838099, 0.0741071}, {0.0910847, 0.0767857},
 {0.0985022, 0.0794643}, {0.1, 0.08}, {0.1, 0.08}, {0.1, 0.08}}
```

- Compare the efficient frontiers of $\mathcal{M}_1$ and $\mathcal{M}_2$.

In[•]:=
```
ListLinePlot[{mnEffFront, mnEffFrontNoShort},
 PlotStyle → {Blue, Red}, PlotLegends → {"Shorts", "NoShorts"}]
```

Out[•]=



Question 3

Use FinancialData[ ] to download the data required to complete the following:

- Secure the closing price data for Microsoft and Apple to 2019-06-30.
- Plot the closing price of Microsoft and Apple on the same graph.
- Compute the daily returns for Microsoft and Apple and plot them on separate graphs.

- Generate a table which contains for each stock in the Dow Jones Industrial index: the ticker symbol, name, and market capitalization.

Solution

In[∘]:=
```
mnMSFT = FinancialData["MSFT", {{2018, 1, 1}, {2018, 9, 29}}, Method → "Legacy"];
mnAAPL = FinancialData["AAPL", {{2018, 1, 1}, {2018, 9, 29}}, Method → "Legacy"];
```

In[∘]:=
```
DateListPlot[{mnMSFT, mnAAPL},
 PlotLegends → {"MFST", "AAPL"}, PlotLabel → "Closing Prices 2018"]
```

Out[∘]=

In[∘]:=
```
mnMSFTRet = Transpose@{Rest[mnMSFT[[All, 1]]], (Rest[#] - Most[#])/Most[#] &[mnMSFT[[All, 2]]]};
mnAAPLRet = Transpose@{Rest[mnAAPL[[All, 1]]], (Rest[#] - Most[#])/Most[#] &[mnAAPL[[All, 2]]]};
```

In[∘]:=
```
DateListPlot[mnAAPLRet, PlotRange → All, PlotLabel → "AAPL 2018 Returns"]
```

Out[∘]=



In[∘]:= ListPlot[{mnMSFTRet⟦All, 2⟧, mnAAPLRet⟦All, 2⟧}ᵀ, AxesLabel → {"MSFT", "AAPL"}, PlotLabel → "Return Comparison 2018", PlotRange → All]

Out[∘]=



In[∘]:= vsDJITickers = FinancialData["^DJI", "Members"]

Out[∘]= {NYSE:MMM, NYSE:AXP, NASDAQ:AAPL, NYSE:BA, NYSE:CAT, NYSE:CVX, NASDAQ:CSCO, NYSE:KO, NYSE:DIS, NYSE:DOW, NYSE:XOM, NYSE:GS, NYSE:HD, NASDAQ:INTC, NYSE:IBM, NYSE:JNJ, NYSE:JPM, NYSE:MCD, NYSE:MRK, NASDAQ:MSFT, NYSE:NKE, NYSE:PFE, NYSE:PG, NYSE:RTX, NYSE:TRV, NYSE:UNH, NYSE:VZ, NYSE:V, NASDAQ:WBA, NYSE:WMT}

In[∘]:= vnDJIMktCap = FinancialData[#, "MarketCap"] & /@ vsDJITickers

Out[∘]= {\$9.76641 × 10^{10}, \$8.32859 × 10^{10}, \$1.82723 × 10^{12}, \$9.09553 × 10^{10}, \$8.25202 × 10^{10}, \$1.46041 × 10^{11}, \$1.68533 × 10^{11}, \$2.16705 × 10^{11}, \$2.32443 × 10^{11}, \$3.73303 × 10^{10}, \$1.57248 × 10^{11}, \$6.70467 × 10^{10}, \$2.9623 × 10^{11}, \$2.12182 × 10^{11}, \$1.09327 × 10^{11}, \$3.92765 × 10^{11}, \$2.99732 × 10^{11}, \$1.63903 × 10^{11}, \$2.17034 × 10^{11}, \$1.51648 × 10^{12}, \$1.78857 × 10^{11}, \$2.03549 × 10^{11}, \$3.41999 × 10^{11}, \$9.52493 × 10^{10}, \$2.82587 × 10^{10}, \$2.92722 × 10^{11}, \$2.49732 × 10^{11}, \$4.31127 × 10^{11}, \$3.20011 × 10^{10}, \$3.83378 × 10^{11}}

In[∘]:= vsDJINames = FinancialData[#, "Name"] & /@ vsDJITickers

Out[∘]= {3M, American Express, Apple, Boeing, Caterpillar, Chevron, Cisco, Coca-Cola, Disney, Dow, Exxon Mobil, Goldman Sachs, Home Depot, Intel, IBM, Johnson & Johnson, JPMorgan Chase, McDonald's, Merck & Co., Microsoft, Nike, Pfizer, Procter & Gamble, Raytheon Technologies Corp, Travelers, UnitedHealth, Verizon Communications, Visa, Walgreens Boots Alliance Inc, Wal-Mart Stores}

In[∘]:= TableForm[{vsDJITickers, vsDJINames, vnDJIMktCap}ᵀ, TableHeadings → {Automatic, {"Symbol", "Name", "Mkt Cap"}}]

Out[•]//TableForm=

	Symbol	Name	Mkt Cap
1	NYSE:MMM	3M	$\$9.76641 \times 10^{10}$
2	NYSE:AXP	American Express	$\$8.32859 \times 10^{10}$
3	NASDAQ:AAPL	Apple	$\$1.82723 \times 10^{12}$
4	NYSE:BA	Boeing	$\$9.09553 \times 10^{10}$
5	NYSE:CAT	Caterpillar	$\$8.25202 \times 10^{10}$
6	NYSE:CVX	Chevron	$\$1.46041 \times 10^{11}$
7	NASDAQ:CSCO	Cisco	$\$1.68533 \times 10^{11}$
8	NYSE:KO	Coca-Cola	$\$2.16705 \times 10^{11}$
9	NYSE:DIS	Disney	$\$2.32443 \times 10^{11}$
10	NYSE:DOW	Dow	$\$3.73303 \times 10^{10}$
11	NYSE:XOM	Exxon Mobil	$\$1.57248 \times 10^{11}$
12	NYSE:GS	Goldman Sachs	$\$6.70467 \times 10^{10}$
13	NYSE:HD	Home Depot	$\$2.9623 \times 10^{11}$
14	NASDAQ:INTC	Intel	$\$2.12182 \times 10^{11}$
15	NYSE:IBM	IBM	$\$1.09327 \times 10^{11}$
16	NYSE:JNJ	Johnson & Johnson	$\$3.92765 \times 10^{11}$
17	NYSE:JPM	JPMorgan Chase	$\$2.99732 \times 10^{11}$
18	NYSE:MCD	McDonald's	$\$1.63903 \times 10^{11}$
19	NYSE:MRK	Merck & Co.	$\$2.17034 \times 10^{11}$
20	NASDAQ:MSFT	Microsoft	$\$1.51648 \times 10^{12}$
21	NYSE:NKE	Nike	$\$1.78857 \times 10^{11}$
22	NYSE:PFE	Pfizer	$\$2.03549 \times 10^{11}$
23	NYSE:PG	Procter & Gamble	$\$3.41999 \times 10^{11}$
24	NYSE:RTX	Raytheon Technologies Corp	$\$9.52493 \times 10^{10}$
25	NYSE:TRV	Travelers	$\$2.82587 \times 10^{10}$
26	NYSE:UNH	UnitedHealth	$\$2.92722 \times 10^{11}$
27	NYSE:VZ	Verizon Communications	$\$2.49732 \times 10^{11}$
28	NYSE:V	Visa	$\$4.31127 \times 10^{11}$
29	NASDAQ:WBA	Walgreens Boots Alliance Inc	$\$3.20011 \times 10^{10}$
30	NYSE:WMT	Wal-Mart Stores	$\$3.83378 \times 10^{11}$

Question 4

Recently, yields on the sovereign debt of several European countries have turned negative. In effect, bond holders are paying these governments for holding their capital.

Yields on German Bunds can be found at

https://www.bloomberg.com/markets/rates-bonds/government-bonds/germany.

Recent data for late September 2019 were {term, yield}:

In[•]:=
```
mnBundYields = {{2, -0.0076}, {5, -0.007}, {10, -0.0059}, {30, -0.0012}};
```

A reasonable estimate for the yield curve can be found using a second-order spline interpolation.

Consider an older Bund with a 10-year term, semi-annual payments, a face amount of €10,000 and a coupon rate of 0.002%. The bond is 3 years into its term. What is the current market price of the bond?

Solution

In[•]:=
```
xBundYield = Interpolation[mnBundYields, Method → "Spline", InterpolationOrder → 2];
```

In[•]:=
```
Show[
 ListPlot[mnBundYields, PlotStyle → {PointSize[Large]}],
 Plot[xBundYield[t], {t, 2, 30}, PlotStyle → {Dashed}],
 PlotLabel → "Bund Yields - Sep 2019",
 AxesLabel → {"Term", "Yield"},
 ImageSize → 500
]
```

Out[•]=



The bond can be viewed as a 7-year bond:

In[•]:=
```
nBundYield7 = xBundYield[7]
```
Out[•]=
```
-0.00657126
```
In[•]:=
```
FinancialBond[{"FaceValue" → 10 000, "Coupon" → 0.002, "Maturity" → 7,
  "CouponInterval" → 1/2}, {"InterestRate" → nBundYield7, "Settlement" → 0}]
```
Out[•]=
```
10 615.
```
Alternately, we could use TimeValue[ ]:

In[•]:=
```
nCoupon = (10 000 × 0.002) / 2
```
Out[•]=
```
10.
```
In[•]:=
```
TimeValue[Annuity[{nCoupon, {0, 10 000}}, 7, 1/2],
 EffectiveInterest[nBundYield7, 1/2], 0]
```
Out[•]=
```
10 615.
```
The bond can equivalently by viewed as a 10-year bond 3 years into its term (with, of course, the appropriate 7-year yield):

In[•]:=
```
FinancialBond[{"FaceValue" → 10 000, "Coupon" → 0.002, "Maturity" → 10,
  "CouponInterval" → 1/2}, {"InterestRate" → nBundYield7, "Settlement" → 3}]
```
Out[•]=
```
10 615.
```